# $Id: PKGBUILD 140196 2011-10-09 08:27:16Z angvp $
# Maintainer: Angel Velasquez <angvp@archlinux.org> 
# Contributor: Sergej Pupykin <sergej@aur.archlinux.org>
# Contributor: William Rea <sillywilly@gmail.com>

pkgname=python-m2crypto
pkgver=0.21.1
pkgrel=1
pkgdesc="A crypto and SSL toolkit for Python"
arch=('i686' 'x86_64')
url="http://wiki.osafoundation.org/bin/view/Projects/MeTooCrypto"
license=('BSD')
depends=('python2' 'openssl')
makedepends=('swig')
source=("http://pypi.python.org/packages/source/M/M2Crypto/M2Crypto-${pkgver}.tar.gz")
md5sums=('f93d8462ff7646397a9f77a2fe602d17')

build() {
  cd "${srcdir}/M2Crypto-${pkgver}"
  python2 setup.py build
}

package() {
  cd "${srcdir}/M2Crypto-${pkgver}"
  python2 setup.py install --root="${pkgdir}/" --optimize=1

  install -D -m644 LICENCE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
